PB192017

IC bureau of mines information circular 8470

COMPUTER PROGRAM FOR PLOTTING DATA POINTS FROM A MULTICHANNEL ANALYZER VIA A REMOTE TELETYPE

Reproduced by the
CLEARINGHOUSE
for Federal Scientific & Technical
Information Springfield Va. 22151

UNITED STATES DEPARTMENT OF THE INTERIOR

BUREAU OF MINES

May 1970

18

COMPUTER PROGRAM FOR PLOTTING DATA POINTS FROM A MULTICHANNEL ANALYZER VIA A REMOTE TELETYPE

By Bruce D. Sartwell, Paul B. Needham, Jr., and Richard S. Graham

* * * * * * * * * * * information circular 8470

UNITED STATES DEPARTMENT OF THE INTERIOR

BUREAU OF MINES

This publication has been cataloged as follows:

Sartwell, Bruce D

Computer program for plotting data points from a multichannel analyzer via a remote teletype, by Bruce D. Sartwell, Paul B. Needham, Jr., and Richard S. Graham. [Washington] U.S. Dept. of the Interior, Bureau of Mines [1970]

14 p. illus. (U.S. Bureau of Mines. Information circular 8470)

1. Information storage and retrieval systems—Computer programs. I. Needham, Paul B., jt. auth. II. Graham, Richard S., jt. auth. III. Title. (Series)

TN23.U71 no. 8470 622.06173

U.S. Dept. of the Int. Library

CONTENTS

	Page
Abstract	1
Introduction	1
Program	2
Examples	5
Appendix	11

ILLUSTRATIONS

1. Data printout for the X-ray energy spectrum obtained from an Fe^{55} source placed in front of a gas-flow proportional counter.......... 6
2. Computer plot of the Mn Kα-line and argon escape peak............... 7
3. Expanded plot of Mn Kα-line.. 8
4. Expanded plot of argon escape peak................................... 8
5. Computer responses and information typed in for integrating the Mn Kα-line and argon escape peaks.................................... 9
6. Computer plot of uranium M-shell and N-shell spectra excited by 120 kev protons.. 10

BLANK PAGE

COMPUTER PROGRAM FOR PLOTTING DATA POINTS FROM A MULTICHANNEL ANALYZER VIA A REMOTE TELETYPE

by

Bruce D. Sartwell,[1] Paul B. Needham, Jr.,[2] and Richard S. Graham[3]

ABSTRACT

A program was written which utilizes a remote teletype input to a time-sharing computer to plot charged particle or X-ray energy spectra which are stored in the memory of multichannel analyzers. The program, written in FORTRAN V, accepts the data printed out by the pulse height analyzer on paper tape and then, after reading these data tapes into the computer, plots on the teletype the desired energy spectra on a scale automatically adjusted for the maximum number occurring in the data.

The program is flexible enough so that the researcher can subsequently expand or contract the vertical (intensity) or horizontal (energy) scales. With a slight modification, the program will accept data printouts from scalers or analog to digital converters. The program also includes a feature for integrating the energy peaks obtained by pulse height analysis.

INTRODUCTION

One of the problems associated with the use of a multichannel pulse height analyzer is "saving" the memory which is displayed on the oscilloscope. The total number of counts accumulated in each channel can be printed out on a teletype, but this does not solve the problem of obtaining a plotted record of the spectrum. The most widely used instrument for obtaining this spectrum is the X-Y recorder which will plot each channel on a relative scale. However, this instrument is expensive (typical cost is $3,000) and when not available, another method for plotting the data is desired.

It is possible to use a computer for plotting data retrieved on punch tape from the pulse height analyzer. The actual curves are plotted by a peripheral computer device known as a Cal-comp[4] plotter. However, the

[1] Physicist, College Park Metallurgy Research Center, Bureau of Mines, College Park, Md.

[2] Research physicist, College Park Metallurgy Research Center, Bureau of Mines, College Park, Md.

[3] Department of Physics, University of Maryland, College Park, Md.

[4] Reference to trade names is made for identification only and does not imply endorsement by the Bureau of Mines.

conversion of the paper punch tape containing the data to a form that can be used by the plotter is very complicated. First, the paper tape and a card conversion program punched on computer punchcards must be submitted through batch processing. Once the data have been converted onto punchcards, the cards must be resubmitted with a plotting program in order that the data can be stored on a magnetic tape in the proper format. Finally, the magnetic tape is mounted on the Cal-comp plotter and the plot is made. For large quantities of data plots, this method is impractical both from the standpoint of the time involved in processing as well as the economical aspects. (The typical cost per plot is $4.) However, with the advent of computers which allow the researcher to submit programs and data through a teletype located at a remote station, the plotting can be performed quickly and easily.

To use this capability of the newer computers, a program was written in FORTRAN V and punched on paper tape for use with a teletype connected to a time-sharing computer. The cost of writing the program and using the computer for plotting is (typical cost per plot is $0.25) many times less than the expense incurred in buying and using an X-Y recorder.

The program also permits operations not possible with the X-Y recorder such as integration of peaks obtained in pulse height analysis and expansion of the horizontal scale so that only certain channels will be plotted.

The program has been used extensively in the authors' laboratory with the analyzer in both pulse height analysis mode and multiscaling mode. In pulse height analysis mode, the program has plotted the spectra as intensity versus energy, and peaks have been integrated to obtain selected spectral intensities. The multiscaling mode has been used with an X-ray spectrometer. In this mode of operation, the plots were of intensity versus degrees theta, the diffraction angle of the spectrometer with the pulses that drive the spectrometer motor used as an external clock to the analyzer. The ability of the program to expand small peaks has also proven very useful in this mode since only one scan of the spectrometer is necessary to observe all the peaks, both low and high intensity, where two or three scans were previously required.

PROGRAM

A listing of the program is found in the appendix. The program was written in FORTRAN V, tested and utilized on the University of Maryland UNIVAC 1108 computer via a teletype input located in the control room of the authors' laboratory. The program was punched on paper tape to eliminate the process of having to type the program by hand each time it is used. The time required to read the tape into the computer is approximately 3 minutes.

After the program is entered into the computer, the computer is then commanded to execute. The first item printed by the computer is a request for the lowest channel, the maximum channel, and the channels per line desired by the experimenter. To plot each datum point (for example, from a 512-channel pulse height analyzer memory), the procedure is to type the following: 001 512 001. If, instead, only every fourth channel is to be plotted, 001 512 004, would be typed.

The program assumes that the first number read into the computer (exclusive of channel identification numbers supplied by the pulse height analyzer) will correspond to channel one, since it is impossible to determine position of numbers on paper tape.

If the user knew in advance that only part of a tape was to be plotted, he could specify the first and last channels to be plotted as long as the last channel number requested was less than or equal to the last channel read into the computer. For example, if a tape contains 512 channels of memory and only channels 100 to 300 were to be plotted, the first items typed into the computer would be 100 300 001.

Once this information has been typed in, the memory data tape from the multichannel analyzer can then be fed into the computer. The present program allows for the channel identification numbers to be printed out for every 10th channel on the horizontal scale. The program recognizes these identification numbers during the execution of the plot routine. The format for accepting the numbers which represent the actual data is 10 six-digit numbers per line with one space between each number.

The first 20 channels of a typical printout of the memory data tape are shown below:

```
000000

000003 000005 000002 000006 000007 000004 000034 000006 000004 000045

000010

000023 000005 000004 000034 000007 000006 000062 000004 000540 000002

000020
```

For pulse height analyzers that do not print an identification number for every 10th channel on the paper tape, or for using other digital readout devices, the program can be easily modified to accept a different format. The only required change is to modify line 8 of the program to read:

```
8      FORMAT ((10(I6,1X))).
```

As soon as the total number of channels specified by the GREATEST CHANNEL is read in, the program will begin plotting. With reference to the plot, the X-axis is the axis representing channel numbers, and the Y-axis is the axis representing intensity. During program execution, the maximum count occurring in the data is determined, and the Y-axis maximum is automatically adjusted so that this maximum is at the top of the Y-axis. The following information is printed out:

(a) the lowest channel number.

(b) the greatest channel number.

(c) the maximum count.

(d) the counts/space on the Y-axis.

(e) the data are plotted as asterisks with the channel number printed each 10th line on the X-axis.

After the plot has been completed, the computer prints out the following message:

TYPE '1' TO CHANGE SCALES.

If either the X- or Y-axis is to be changed, the number "1" is typed. The computer responds with

LOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/LINE, MAXIMUM COUNT.

The new parameters are then typed in the same manner as at the beginning of the program execution. The experimenter has wide latitude in plotting specific sections of the analyzer memory, plotting every second, fourth, eighth, etc. channel, or expanding the Y-axis to obtain more detail (for example, for smaller peaks in either pulse height analysis or multiscaling mode). The Y-axis is scaled by what is typed in for the maximum count. Channels that have a higher number of counts than the specified maximum will be included in the plot with the statement:

OFF SCALE: CHANNEL XXX, COUNT = YYYYYY.

Thus, the channel numbers and the counts in these off-scale channels will be known.

If one does not wish to change scales, any one-digit number except "1" should be typed. The computer responds with

TYPE '2' TO INTEGRATE BETWEEN TWO CHANNELS.

If it is desired to obtain the sum of the counts between two channels or integrate a peak obtained from pulse height analysis, a "2" should be typed into the computer. The computer will respond with:

START CHANNEL, STOP CHANNEL.

The user then types in the first channel and the last channel of the region to be integrated using the following format:

XXX YYY.

Once this has been typed in, the computer will sum the counts in the specified spectral region and print out the result.

The computer will then repeat the message:

TYPE '2' TO INTEGRATE BETWEEN TWO CHANNELS.

If another peak is to be integrated, the experimenter must type in the number "2" once again. Otherwise, any other one-digit number will cause the program to terminate.

EXAMPLES

To illustrate the procedure discussed in the preceding sections, an X-ray energy spectrum was obtained from a Fe^{55} source placed in front of a gas-flow proportional counter. The spectrum was accumulated in 256 channels of memory on an ND 2201 Multichannel Pulse Height Analyzer.

Figure 1 shows the format of the numbers and the actual number of counts accumulated in each channel. A paper tape copy of the data was made for use with the plot program.

Figure 2 shows the computer responses and the first plot. Every fourth channel was plotted so that 64 total points were plotted.

The number of channels decided upon by the experimenter is generally determined by the energy resolution of the peaks in the energy spectra to be plotted. Thus, for low-resolution detectors such as proportional counters, not all channels are required for studying the spectra, as opposed to the high-resolution solid state detectors which generally require every channel to be plotted. The first line in figure 3 is the computer response after the first plot (fig. 2) was made. A "1" was typed in to change scales. The Mn $K\alpha$ peak and the argon escape peak were plotted in figure 2. When this had been completed, an expanded version of Mn $K\alpha$ peak was plotted and each channel plotted instead of every fourth channel. The line "150 230 001 003543" was typed in to request the Mn $K\alpha$ peak shown in figure 3.

Next, the argon escape peak was expanded. To accomplish this, "1" was typed in again. The maximum count was readjusted to move the top of the argon peak to approximately one-half full scale. A part of the Mn $K\alpha$ peak was also included to demonstrate the feature whereby all off-scale points are indicated with the channel number and the count. This plot is shown in figure 4. Finally, both the Mn $K\alpha$ and the argon escape peaks were individually integrated. Figure 5 shows the computer responses and the information that was typed in by the experimenter. He had to type in the lowest and highest channel number around each peak to obtain a correct integration. To terminate the run, a "3" was typed in at the end.

A final example is presented which utilizes a different X-ray spectrum accumulated in the memory of the pulse height analyzer. Figure 6 shows an X-ray energy spectrum obtained from bombardment of a thick uranium target by 120 kev protons. The peak in channel 46 is a composite of several uranium N-shell transitions and the peak in channel 88 is a composite of several M-shell transitions. The spectrum was obtained in 128 channels of analyzer

```
000000
000100 000000 000000 000000 000000 000000 000000 000000 000000 000000
000010
000000 001772 002137 000519 000341 000288 000243 000217 000239 000217
000020
000187 000199 000205 000229 000198 000207 000166 000163 000157 000147
000030
000135 000139 000145 000135 000150 000125 000132 000143 000150 000148
000040
000099 000129 000118 000122 000121 000126 000126 000129 000125 000112
000050
000126 000135 000137 000151 000121 000128 000123 000118 000125 000121
000060
000128 000139 000114 000143 000127 000135 000122 000127 000116 000140
000070
000133 000134 000121 000126 000110 000131 000133 000150 000139 000121
000080
000163 000143 000155 000155 000156 000148 000164 000206 000195 000202
000090
000238 000223 000262 000276 000341 000356 000350 000373 000419 000397
000100
000422 000481 000500 000444 000497 000534 000497 000495 000501 000510
000110
000522 000498 000487 000467 000416 000453 000421 000383 000405 000374
000120
000340 000292 000307 000305 000295 000283 000280 000269 000254 000259
000130
000227 000250 000216 000230 000213 000246 000214 000229 000238 000228
000140
000242 000263 000234 000257 000279 000269 000287 000299 000291 000357
000150
000403 000427 000475 000472 000500 000530 000638 000691 000792 000792
000160
000865 001011 001083 001258 001337 001457 001609 001705 001860 002068
000170
002137 002267 002504 002608 002676 002899 002911 002937 003221 003299
000180
003314 003340 003420 003506 003495 003543 003449 003364 003328 003221
000190
003244 003094 002943 002804 002832 002617 002513 002396 002225 002056
000200
001884 001817 001642 001494 001372 001316 001207 001063 000906 000895
000210
000760 000759 000654 000584 000516 000464 000438 000356 000298 000287
000220
000218 000219 000176 000196 000157 000141 000121 000088 000091 000075
000230
000060 000049 000052 000056 000037 000027 000037 000032 000026 000022
000240
000027 000021 000017 000010 000021 000011 000010 000012 000012 000010
000250
000006 000009 000007 000010 000004 000007
```

FIGURE 1. - Data Printout for the X-Ray Energy Spectrum Obtained From an Fe^{55} Source Placed in Front of a Gas-Flow Proportional Counter.

```
LOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/SPACE
001 256 004

TOP CHANNEL =    1       BOTTOM CHANNEL = 256
MAXIMUM COUNT=    3543   COUNTS/SPACE=      59.050
```

FIGURE 2. - Computer Plot of the Mn Kα-Line and Argon Escape Peak.

```
TYPE '1' TO CHANGE SCALES
1
LOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/LINE, MAXIMUM COUNT
150 230 001 003543
TOP CHANNEL = 150      BOTTOM CHANNEL = 230
MAXIMUM COUNT=   3543   COUNTS/SPACE=      59.050
```



FIGURE 3. - Expanded Plot of Mn Kα-Line.

```
TYPE '1' TO CHANGE SCALES
1
LOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/LINE, MAXIMUM COUNT
080 170 001 001000
TOP CHANNEL =  80      BOTTOM CHANNEL = 170
MAXIMUM COUNT=   1000   COUNTS/SPACE=      16.667
```




```
OFF SCALE: CHANNEL 162  COUNT   1011
OFF SCALE: CHANNEL 163  COUNT   1083
OFF SCALE: CHANNEL 164  COUNT   1258
OFF SCALE: CHANNEL 165  COUNT   1337
OFF SCALE: CHANNEL 166  COUNT   1457
OFF SCALE: CHANNEL 167  COUNT   1609
OFF SCALE: CHANNEL 168  COUNT   1705
OFF SCALE: CHANNEL 169  COUNT   1860
OFF SCALE: CHANNEL 170  COUNT   2068
```

FIGURE 4. - Expanded Plot of Argon Escape Peak.

```
 TYPE '1' TO CHANGE SCALES
2

 TYPE '2' TO INTEGRATE BETWEEN TWO CHANNELS
2

 START CHANNEL, STOP CHANNEL
150 220

 THE INTEGRATED CHANNELS =  127205


 TYPE '2' TO INTEGRATE BETWEEN TWO CHANNELS
2

 START CHANNEL, STOP CHANNEL
086 130

 THE INTEGRATED CHANNELS =   16341

 TYPE '2' TO INTEGRATE BETWEEN TWO CHANNELS
3


N
 ORMAL EXIT.  EXECUTION TIME:       1845   MILLISECONDS.
```

FIGURE 5. - Computer Responses and Information Typed in for Integrating the Mn Kα-Line and Argon Escape Peaks.

memory, but the first 14 channels contained background counts that were much greater than the peaks themselves. If these had been included in the plot, the size of the peaks would have been greatly reduced and little observable detail would have been obtained. The procedure used to avoid this was to type in for LOWEST CHANNEL, channel 15. Channels 111 to 128 contained very few counts and were omitted by specifying the GREATEST CHANNEL as channel 110. The maximum count from channels 15 to 110 was found when the 128 channels of memory were read in from the paper tape, and the Y-axis was adjusted for that maximum.

FIGURE 6. - Computer Plot of Uranium M-Shell and N-Shell Spectra Excited by 120 kev Protons.

APPENDIX

The complete computer program, written in FORTRAN V, appears at the end of this appendix. The control statements that were used on the UNIVAC 1108 are also included for reference purposes. After the control statement #ELT, IS TYPLOT is typed in, the usual procedure for using the program is to read it into the computer from paper tape. While this is done, the program is not printed by the teletype. Thus, the control statement, #FOR, U TYPLOT,PLOT, which calls for the computer to compile the program, would appear directly after the control statement given above. The control statement #XQT calls for the program to execute, and the computer will type LOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/LINE.

In order for the program to be more easily understood, the symbols used in the program and their meanings are listed below.

SYMBOL	MEANING
LCHNL	The first channel to be plotted.
MCHNL	The last channel to be plotted.
N	The number of channels per line to be plotted; that is, whether every channel, every second, every fourth, etc. channel is to be plotted.
J(LL)	The total number of counts contained in channel number LL.
MCOUNT	The greatest number of counts in any one channel that is to be plotted.
XQ5	The number of counts per space on the Y-axis.
NY	The space on the Y-axis where the asterisk representing the data point is to be plotted.
A(K)	The K'th position on the Y-axis. Thus the statement A(NY)=AST means that on the NY'th space on the Y-axis, an asterisk will be typed.
M6	This variable keeps track of where on the X-axis the plot is and provides for typing the channel number every 10th space.
AST,BLANK,PER,XMIN	These are the alphabetic representations of the possible characters that can be typed on each line. The actual typed characters are found in line 1 of the program.

SYMBOL	MEANING
NSTART,NSTOP	These are the channels representing the limits of integration of a portion of the spectrum.
NSUM	This is the total number of counts between channels NSTART and NSTOP.

```
@RUN PLOT,750-06-002,TELETYPE
@ELT,IS TYPLOT
ELT PROCESSOR LEVEL     3
C       THIS PROGRAM ALLOWS ONE TO PLOT SPECTRA CONTAINED IN THE
C       MEMORY OF MULTICHANNEL ANALYZERS AND TO INTEGRATE PEAKS.
        DATA AST/'*'/BLANK/' '/PER/'     .'/XMIN/'-'/
        DIMENSION J(512),A(62)
        WRITE (6,4)
4       FORMAT ('OLOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/SPACE ')
        READ (5,6) LCHNL,MCHNL,N
6       FORMAT (I3,1X,I3,1X,I3)
        READ (5,8) (J(K),K=1,MCHNL)
8       FORMAT ((/10(I6,1X)))
9       MCOUNT=J(LCHNL)
        DO 11 K=LCHNL,MCHNL
11      IF (J(K).GT.MCOUNT) MCOUNT=J(K)
        GO TO 17
13      WRITE (6,14)
14      FORMAT ('OLOWEST CHANNEL, GREATEST CHANNEL, CHANNELS/LINE, ',
       *'MAXIMUM COUNT  ')
        READ (5,16) LCHNL, MCHNL,N,MCOUNT
16      FORMAT (3(I3,1X),I6)
17      COUNT=MCOUNT
        XQ5=COUNT/60.
        M6=0
        DO 45 LL=LCHNL,MCHNL,N
        X=J(LL)
        NY=X/XQ5+2.
        IF (NY.GT.62.OR.X.GT.COUNT.OR.X.LT.0.) GO TO 39
        DO 25 I=2,62
25      A(I)=BLANK
        A(NY)=AST
        IF (M6.EQ.0) GO TO 42
        IF (M6.EQ.20) GO TO 33
        A(1)=PER
        WRITE (6,31) (A(I),I=1,62)
31      FORMAT (A5,61A1)
        GO TO 37
33      A(1)=XMIN
        WRITE (6,35) LL,(A(I),I=1,62)
35      FORMAT (1X,I3,62A1)
        M6=10
37      M6=M6+1
        GO TO 45
39      WRITE (6,40) LL,J(LL)
40      FORMAT (' OFF SCALE: CHANNEL ',I3,'  COUNT ',I6)
        GO TO 37
42      WRITE (6,43) LCHNL,MCHNL,MCOUNT,XQ5
43      FORMAT ('OTOP CHANNEL = ',I3,6X,'BOTTOM CHANNEL = ',I3/
       *' MAXIMUM COUNT= ',I6,'   COUNTS/SPACE= ',F10.3/
       *'      !.........!.........!.........!.........!.........!',
       *'.........!')
        GO TO 33
45      CONTINUE
```

```
      WRITE (6,47)
47    FORMAT ('OTYPE ''1'' TO CHANGE SCALES')
      READ (5,49) I
49    FORMAT (I1)
      IF (I.EQ.1) GO TO 13
51    WRITE (6,52)
52    FORMAT ('OTYPE ''2'' TO INTEGRATE BETWEEN TWO CHANNELS')
      READ (5,54) MM
54    FORMAT (I1)
      IF (MM.NE.2) GO TO 67
      WRITE (6,57)
57    FORMAT ('OSTART CHANNEL, STOP CHANNEL  ')
      READ (5,59) NSTART, NSTOP
59    FORMAT (I3,1X,I3)
      NSUM=0
      DO 63 NN=NSTART,NSTOP
      NSUM=NSUM + J(NN)
63    CONTINUE
      WRITE (6,65) NSUM
65    FORMAT ('OTHE INTEGRATED CHANNELS = ',I7)
      GO TO 51
67    STOP
      END
@FOR,U TYPLOT,PLOT
@MAP,N
@XQT
```